Maple-assisted derivation of recurrence for A190027

Robert Israel

05 Mar 2023

There are $2^{15} = 32768$ configurations for a 3×5 sub-array.

We encode these configurations as lists in the order $\begin{bmatrix} x_1 & x_2 & x_3 & x_4 & x_5 \\ x_6 & x_7 & x_8 & x_9 & x_{10} \\ x_{11} & x_{12} & x_{13} & x_{14} & x_{15} \end{bmatrix}$ and enumerate them.

```
> Configs:= [seq(convert(n,base,2)[1..15],n=2^15..2^16-1)]:
```

However, not all of these configurations are allowed, because each 3×3 submatrix must commute with its neighbours.

```
> with(LinearAlgebra):
  subm:= proc(x)  [Matrix(3,3,[x[1],x[2],x[3],x[6],x[7],x[8],x[11],x
  [12],x[13]]),
  Matrix(3,3,[x[2],x[3],x[4],x[7],x[8],x[9],x[12],x[13],x[14]]),
  Matrix(3,3,[x[3],x[4],x[5],x[8],x[9],x[10],x[13],x[14],x[15]])]
  end proc:
> Configs:= select(proc(x) local M,Z; M:= subm(x); Z:= Matrix(3,3);
  Equal(M[1].M[2]-M[2].M[1],Z) and Equal(M[2].M[3]-M[3].M[2],Z) end
  proc, Configs):
  nops(Configs);
```

$$228 \qquad (1)$$

There are only 228 allowed configurations.
Consider the 228×228 transition matrix T with entries $T_{ij} = 1$ if the first three rows of a 4×5 sub-array could be in configuration i while the last three rows are in configuration j, and 0 otherwise. The following code computes it.

```
> Compatible:= proc(i,j) local Xi,Xj,Mi,Mj,Z;
   Xi:= Configs[i]; Xj:= Configs[j];
   if Xi[6..15] <> Xj[1..10] then return 0 fi;
   Mi:= subm(Xi);
   Mj:= subm(Xj);
   Z:= Matrix(3,3);
   if Equal(Mi[1] . Mj[1] - Mj[1] . Mi[1], Z) and
   Equal(Mi[2] . Mj[2] - Mj[2] . Mi[2], Z) and
   Equal(Mi[3] . Mj[3] - Mj[3] . Mi[3], Z) then 1 else 0 fi
  end proc:
  T:= Matrix(228,228,storage=sparse,Compatible):
```

Let u be the column vector of 228 ones.. Then we should have $a(n) = u^T T^{n-1} u$ for all n. Here are the first few.

```
> u:= Vector(228,1):
> Tu[0]:= u:
  for i from 1 to 20 do Tu[i]:= T . Tu[i-1] od:
> seq(u^%T . Tu[i], i=0..20);
```

228, 79, 158, 274, 362, 430, 506, 610, 790, 1094, 1539, 2128, 2917, 3951, 5309, 7171, 9723, (2) 13188, 17946, 24428, 33181

The recurrence shows up as a linear dependence among $T^n u$. We gather these as columns of a matrix L, and stop when it has less than full column rank.

```
> L:= u:
  for nn from 1 do
    L:= <L|Tu[nn]>;
    if LinearAlgebra:-Rank(L) < nn+1 then printf("Success at n=
  %d\n",nn); break fi;
  od:
Success at n=11
```

The recurrence can then be found from the null space of the matrix L.

```
>  P:= convert(LinearAlgebra:-NullSpace(L)[1],list);
```

$$P := [0, 0, 0, 0, 1, -1, -1, 2, -1, 1, -2, 1] \quad \textbf{(3)}$$

That is, $(T^4 - T^5 - T^6 + 2\,T^7 - T^8 + T^9 - 2\,T^{10} + T^{11})\,u = 0$, which implies the recurrence

$$a(n-7) - a(n-6) - a(n-5) + 2\,a(n-4) - a(n-3) + a(n-2) - 2\,a(n-1) + a(n) = 0$$

for $n \geq 12$.